



# PUSHTX and its Building Blocks

White Paper

WP1605

Created: 20/05/2021

Last updated: 14/12/2021

## Copyright

Information in this document is subject to change without notice and is furnished under a license agreement or nondisclosure agreement. The information may only be used or copied in accordance with the terms of those agreements.

All rights reserved. No part of this publication may be reproduced, stored in a retrieval system, or transmitted in any form or any means electronic or mechanical, including photocopying and recording for any purpose other than the purchaser's personal use without the written permission of nChain Licensing AG. The names of actual companies and products mentioned in this document may be trademarks of their respective owners. nChain Licensing AG accepts no responsibility or liability for any errors or inaccuracies that may appear in this documentation.

## Disclaimer

The information contained herein, such as code, data structures, pseudo-code, etc. is provided "as is", no warranty, express or implied, regarding the merchantability or fitness for purpose of the information contained herein, is provided by nChain Licensing AG or any of its group companies. Neither nChain Licensing AG, nor any of its group companies, accepts any liability for any consequential, incidental, special or indirect damages that may occur through use of the information contained herein.

## Authors

---

Name	Title
Wei Zhang	Principal Researcher

---

## Contents

1	PUSHTX and Its Building Blocks .....	4
1.1	Generating the signature in-script.....	4
1.2	Constructing the message in-script.....	5
1.3	Example of a PELS .....	8
1.4	Optimisation.....	12
2	Security Analysis.....	15
3	Appendix .....	18

# 1 PUSHTX and Its Building Blocks

The core idea of PUSHTX (originally invented by Y. Chan and D. Kramer at nChain in 2017) is to generate a signature in-script on a data element on the stack and call OP\_CHECKSIG to verify the signature. If it passes, it implies that the message constructed by OP\_CHECKSIG is identical to the data element pushed to the stack. Therefore, it achieves the effect of pushing the current spending transaction to the stack. Since its adoption by sCrypt, STAS token and Sensible Contracts, the PUSHTX idea has been widely discussed and tested in practice. The purpose of this document is to offer some security insight and optimisations while revisiting PUSHTX with an example of a perpetually enforcing locking script. We focus on two most important building blocks of PUSHTX, the signature generation and the message construction. The size of its spending transaction is intended to be a benchmark for reference. Other techniques of pushing transactions to the stack are out of scope of this white paper.

A perpetually enforcing locking script (PELS) is a locking script that enforces some condition or conditions on all future transactions in the spending chain that originates from the output that contains this locking script. One example to achieve this is to design a locking script that forces the locking script in the spending transaction to be the same as itself. Note that a locking script with an enforcement only on the next spending transaction would have a much simpler design. PELs are particularly useful for the sender (originator) as they can be ensured that all future spending transactions will follow the rules which they set out in the locking script. Any violation of the rules would invalidate the transaction in terms of script execution.

## 1.1 Generating the signature in-script

The first building block is to generate the signature for a given message  $m$ . We assume that the following script segment is part of a locking script, and the input data can be either in an unlocking script or hard coded in the same locking script.

```
[sign]:= OP_HASH256  $k^{-1}$  OP_MUL  $k^{-1}ra$  OP_ADD  $n$  OP_MOD  $r$  [toDER]
SIGHASH_FLAG OP_CAT
```

Input data:  $m$

### Remarks

1. The equation for computing  $s$  in the ECDSA signature is  $k^{-1}(z + ra) \bmod n$ , where  $z$  is the double SHA256 of the message  $m$  to be signed. We write the equation in script as  $(k^{-1}z + k^{-1}ra) \bmod n$  to indicate that  $k^{-1}$  and  $k^{-1}ra$  can be precomputed. It would be very costly to compute modular inverse  $k^{-1} \bmod n$  in script. As we are not using the signature for authenticity, the private key  $a$  and the ephemeral key  $k$  can be chosen at wish and shown publicly.
2. The script segment `[sign]` as part of the locking script needs to fix both the ephemeral key  $k$  and the private key  $a$ . Although anyone can generate a valid signature using `[sign]`, the focus is on the input  $m$ . The requirement is that there is only one value of  $m$  that can pass OP\_CHECKSIG for any given spending transaction. If the private key or the public key is not fixed, then the transaction will be malleable. The detail can be found in Section 2. If the ephemeral key  $k$  is not fixed, then anyone can

use a different  $k$  to create a valid transaction with a different transaction ID, which is not desirable in some use cases.

- To further optimise the script segment, one can choose small values for  $k$  and  $a$  such as  $1$ , and they can be the same every time. Note that if  $k = a = 1$ , then  $s = z + G_x \bmod n$ , where  $G_x$  is the  $x$ -coordinate of the generator point  $G$ . The compressed public key will be  $G_x$  too. The definition of `[sign]` can be re-written as

```
[sign]:= OP_HASH256 G_x OP_ADD n OP_MOD G_x [toDER]
SIGHASH_FLAG OP_CAT
```

- The script segment `[toDER]` is to convert the pair  $(r, s)$  to the canonical DER format. This is the only format accepted by `OP_CHECKSIG`. It forces  $s$  to be in the range between  $0$  and  $n/2$  to avoid transaction ID malleability. Although this is a policy rule in the Bitcoin network, it seems that Bitcoin nodes are unlikely to accept alternatives.
- Note that `SIGHASH_FLAG` in `[sign]` is compulsory as `OP_CHECKSIG` expects it. It can be used to specify which part of the spending transaction should be pushed to the stack. For example, the flag `ALL` would require all the inputs and outputs to be included in the message  $m$ , while `SINGLE|ANYONECANPAY` would require the input corresponding to this locking script and its paired output to be included in  $m$ .
- If we extend the script segment to `"OP_DUP [sign] <PK >"` with input  $m$ , the stack from bottom to top will look like `[m, Sig, PK]` after its execution. A call to `OP_CHECKSIGVERIFY` will consume the signature and the public key, leaving  $m$  on the top of the stack. If the verification is successful, then one can be convinced that the message  $m$  left on the stack is an accurate representation of the spending transaction.

## 1.2 Constructing the message in-script

The signed message in its serialised format is different from the serialised transaction. The latter gives away all the information about the transaction, while the signed message unintentionally conceals some information about the transaction in hash values and offers some information about the output being spent, i.e., its value and its locking script.

The message  $m$  cannot be fully embedded in the locking script as it contains the locking script itself and some unknown information on the future spending transaction. Only some of the fields can be explicitly enforced in the locking script, e.g., version, sequence number, or locktime. The message  $m$  is either provided in the unlocking script in its entirety or constructed in script with some inputs from the unlocking script and instructions from the locking script. We will focus on the latter as it is more restrictive from the perspective of a spending transaction. Note that the goal of constructing the message is to enforce desired values for some data fields in the spending transaction. The table below captures all the data fields in the message and whether they should or can be fixed in the locking script. "Optional" is given if it is use-case specific.

**Table 1** - Components of a signed message

	<b>Items</b>	<b>Fixed explicitly in locking script or not</b>
<b>1</b>	Version 4 bytes little endian	Optional
<b>2</b>	Hash of input outpoints 32 bytes	Infeasible due to circular reference of TxID
<b>3</b>	Hash of input sequence numbers 32 bytes	Optional, recommend “Not” to allow more flexibility in spending transaction
<b>4</b>	Input outpoint 32 bytes + 4 bytes in little endian	Infeasible due to circular reference of TxID (although 4 bytes index can be optional)
<b>5</b>	Length of previous locking script	Optional, recommend “Not” for simplicity
<b>6</b>	Previous locking script	Infeasible due to circular reference of the locking script
<b>7</b>	Value of previous locking script 8 bytes (little endian)	Optional
<b>8</b>	Sequence number 4 bytes (little endian)	Optional
<b>9</b>	Hash of outputs 32 bytes	Optional if it is known before hand, otherwise, infeasible to be fixed.
<b>10</b>	Locktime 4 bytes in little endian	Optional
<b>11</b>	Sighash flag 4 bytes in little endian	Recommend being fixed for more restrictiveness

From now on, the data fields in the table will be referred as item 1, 2, 3, etc.

When it is optional, whether to provide the data in the locking or unlocking script depends on use cases. A general rule is that if the data is available or known at the time of creating the locking script, then they can be included in the locking script. Another aspect to consider is the size of the transaction and its spending transaction. By shifting the data between the locking and unlocking script, one can shift some of the transaction fee cost between the senders of the two transactions.

Note that when we say infeasible due to circular references, the granularity is set at data fields. For example, partial locking script or even a small part of a transaction ID (e.g., fixing the first two bytes of a 32-byte transaction ID and allow iterations through some fields in a serialised transaction) can be fixed in the locking script if required.

As mentioned earlier, although the focus is to construct the message  $m$ , the goal is to use  $m$  to enforce values on different fields in the spending transaction. To enforce the data behind the hash values, i.e., item 9, the locking script should be designed to request the pre-image, hash

them in-script, and then construct the message to be signed in-script. Taking item 9 as an example, to enforce the outputs in the current transaction, we can have

```
[outputsRequest] := OP_DUP OP_HASH256 OP_ROT OP_SWAP OP_CAT <item 10 and 11> OP_CAT
```

```
Input data: <item 1 to 8> <serialised outputs in current transaction>
```

## Remarks

1. The script segment `[outputsRequest]` takes item 1 to 8 and the serialised outputs on the stack to construct item 9, and concatenate with item 10 and 11 to obtain the message  $m$  in-script. By calling `[sign] <Gx> OP_CHECKSIGVERIFY` after `[outputsRequest]` and passing the verification, one can be convinced that the serialised outputs left on the top of the stack is a true representation of the outputs in the current transaction.
2. It is also very useful to leave a copy of `<item 1 to 7>` on the stack for comparison. This can be achieved by modifying the script segment as below.

```
[outputsRequest] := OP_2DUP OP_HASH256 OP_SWAP <item 8> OP_CAT OP_SWAP OP_CAT <item 10 and 11> OP_CAT
```

```
Input data: <item 1 to 7> <serialised outputs in current transaction>
```

After executing the modified `[outputsRequest]` on the input data, we can call `[sign] <Gx> OP_CHECKSIGVERIFY` to consume the message. The stack will have the current serialised outputs on the top followed by `<item 1 to 7>`. We will use the modified script segment later to enforce the output in a spending transaction.

3. It is simpler if consecutive items are grouped together as in `<item 1 to 7>`. They are either all in an unlocking script or all fixed in a locking script. However, a more granular approach is available at a potential cost of having a more complex script.

Note that the serialisation format for current outputs is

- a. value of the output 8 bytes (little endian),
- b. length of the locking script,
- c. the locking script, and
- d. concatenate serialised outputs in order if there is more than one output.

The serialisation format for previous output (item 5 to 7) in a signed message is

- a. length of the locking script,
- b. the locking script, and
- c. value of the output 8 bytes (little endian).

In the following example, we will compare the previous output with the output in the current spending transaction and force them to be identical. The two formats will be useful for designing the locking script for the comparison.

### 1.3 Example of a PELS

Suppose that Alice is a root Certificate Authority (CA) and Bob is a subordinate CA. Alice is going to delegate some work to Bob which would require Bob to publish transactions on-chain as attestations to certificates. Alice does not want Bob to spend the output on anything else. Therefore, Alice is going to force all the subsequent spending transactions to have a fixed `[P2PKH Bob]` locking script and a fixed output value. Bob can spend the output as he can generate valid signatures, but he cannot choose any output other than sending the same amount to himself. For illustration purpose and simplicity, we ignore `OP_RETURN` payload throughout the example.

Alice constructs the initial transaction as shown below.

TxID <sub>0</sub>				
Version	1	Locktime	0	
In-count	1	Out-count	1	
Input list			Output list	
Outpoint	Unlocking script	nSeq	Value	Locking script
<i>Outpoint<sub>A</sub></i>	<code>&lt; Sig<sub>A</sub> &gt;</code> <code>&lt; PK<sub>A</sub> &gt;</code>	FFFFFFFF	1000	<code>[outputsRequest][sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK<sub>B</sub>)&gt; OP_EQUALVERIFY OP_CHECKSIG</code>

Figure 1 - Initial transaction that contains a PELS output

The script segments are defined as:

```
[outputsRequest]:= OP_2DUP OP_HASH256 OP_SWAP <item 8> OP_CAT OP_SWAP
OP_CAT <item 10 and 11> OP_CAT
[sign]:= OP_HASH256 Gx OP_ADD n OP_MOD [toDER] SIGHASH_FLAG OP_CAT
Gcompressed
[toDER]:= [toCanonical][concatenations]
[toCanonical]1:= OP_DUP n/2 OP_GREATERTHAN OP_IF n OP_SWAP OP_SUB OP_ENDIF
[concatenations]:= OP_SIZE OP_DUP <0x24> OP_ADD <0x30> OP_SWAP OP_CAT
<0220||Gx||02> OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
```

Note that “`OP_SWAP <0x68> OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY`” is to extract the previous output from the signed message (verified and left on the top of the stack), then swap the position of the value field and the locking script to form the expected current output, and finally compare it with the actual current output provided in the unlocking script and its integrity is implied by the integrity of the signed message.

The length of the locking script is roughly

<sup>1</sup> This is to make sure that  $s$  is in the range between 0 and  $n/2$ . If  $s > n/2$ , we let  $s = n - s$ .



$$(7 + 12) + (6 + 32 + 32 + 33) + (6 + 32 + 32) + (11) + (15 + 34) + (14 + 20) = 286 = 0x011e.$$

Note that these numbers are not meant to be precise, but they should be accurate enough to be in the right magnitude. For a more accurate result, please refer to Appendix.

To spend the transaction, Bob creates the spending transaction as below.

TxID <sub>1</sub>				
Version	1	Locktime	0	
In-count	1	Out-count	1	
Input list			Output list	
Outpoint	Unlocking script	nSeq	Value	Locking script
TxD <sub>0</sub>   0	<pre>&lt; Sig<sub>B</sub> &gt; &lt; PK<sub>B</sub> &gt; &lt; Data<sub>1</sub> &gt; &lt; Data<sub>2</sub> &gt;</pre>	FFFFFFFF	1000	<pre>[outputsRequest][sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG</pre>

Figure 2 - spending the PELS output

The unlocking script contains a data element *Data<sub>1</sub>* which represents items 1 to 7 and can be written as:

```
010000002268f59280bdb73a24aae224a0b30c1f60b8a386813d63214f86b98261a6b8763
bb13029ce7b1f559ef5e747fcac439f1455a2ec7c5f09b72290795e70665044TxID000000
000011e{[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP <0x68> OP_SPLIT
OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP
OP_HASH160 <H(PK_B)> OP_EQUALVERIFY OP_CHECKSIG}e803000000000000
```

Table 2 - Data element *data1* explained

	Items	Value
1	version	01000000
2	Hash of input outputs	2268f59280bdb73a24aae224a0b30c1f60b8a386813d63214f86b98261a6b876
3	Hash of input sequence numbers	3bb13029ce7b1f559ef5e747fcac439f1455a2ec7c5f09b72290795e70665044
4	Input outpoint	TxD <sub>0</sub> 00000000
5	Length of previous locking script	011e

<b>6</b>	Previous locking script	<pre> {[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG}                     </pre>
<b>7</b>	Value of previous locking script	e803000000000000

The data element *Data<sub>2</sub>* represents the output in *TxID<sub>1</sub>* (value || locking script length || locking script) and can be written as:

```

e803000000000000011e{[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP
<0x68> OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT
OP_EQUALVERIFY OP_DUP OP_HASH160 <H(PK_B)> OP_EQUALVERIFY OP_CHECKSIG}
                    
```

The full script to be executed during the validation of *TxID<sub>1</sub>* is

```

<SigB> <PKB> <Data1> <Data2> [outputsRequest] [sign] OP_CHECKSIGVERIFY
OP_SWAP <0x68> OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT
OP_EQUALVERIFY OP_DUP OP_HASH160 <H(PK_B)> OP_EQUALVERIFY OP_CHECKSIG
                    
```

After the first `OP_CHECKSIGVERIFY`, we will have `<SigB> <PKB> <Data1> <Data2>` on the stack (rightmost on the top).

**Table 3** - Stack execution

Step	The stack	To execute
<b>1</b>	<code>&lt;Sig<sub>B</sub>&gt; &lt;PK<sub>B</sub>&gt; &lt;Data<sub>1</sub>&gt; &lt;Data<sub>2</sub>&gt;</code>	<code>OP_SWAP &lt;0x68&gt;</code>
<b>2</b>	<code>&lt;Sig<sub>B</sub>&gt; &lt;PK<sub>B</sub>&gt; &lt;Data<sub>2</sub>&gt; &lt;Data<sub>1</sub>&gt; &lt;0x68&gt;</code>	<code>OP_SPLIT OP_NIP</code>
<b>3</b>	<code>&lt;Sig<sub>B</sub>&gt; &lt;PK<sub>B</sub>&gt; &lt;Data<sub>2</sub>&gt;</code> <code>&lt;011e</code> <pre> {[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG} e80300000000000000&gt;                     </pre>	<code>OP_SWAP OP_8 OP_SPLIT</code>
<b>4</b>	<code>&lt;Sig<sub>B</sub>&gt; &lt;PK<sub>B</sub>&gt; &lt;Data<sub>2</sub>&gt;</code> <code>&lt;011e</code> <pre> {[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY                     </pre>	<code>OP_SWAP OP_CAT</code>

	<pre> OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG} e803000000000000&gt; &lt;e803000000000000&gt; &lt;011e {[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG}&gt; </pre>	
5	<pre> &lt; Sig<sub>B</sub> &gt; &lt; PK<sub>B</sub> &gt; &lt; Data<sub>2</sub> &gt; &lt;011e {[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG} e803000000000000&gt; &lt;011e {[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP &lt;0x68&gt; OP_SPLIT OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG} e803000000000000&gt; </pre>	OP_EQUALVERIFY
6	<pre> &lt; Sig<sub>B</sub> &gt; &lt; PK<sub>B</sub> &gt; </pre>	<pre> OP_DUP OP_HASH160 &lt;H(PK_B)&gt; OP_EQUALVERIFY OP_CHECKSIG </pre>
7	True	

The size of  $TxD_1$  is

$$\begin{aligned}
 & \text{version} + \text{locktime} + \text{input} + \text{output} \\
 & = 4 + 4 + (36 + 72 + 33 + 104 + 287 + 8 + 287 + 8 + 4) + (8 + 287) \\
 & = 1142 \text{ bytes.}
 \end{aligned}$$

### Transaction Fee

Given the current setting, Bob can add his own input to cover the transaction fee. If Alice uses `SIGHASH_SINGLE|ANYONECANPAY` in the script segment `[sign]`, then Bob can add another output to collect changes. This effectively makes the enforcement from Alice’s locking script perpetual. One can think of this as a node-enforced smart contract between Alice and Bob.

It is also possible for the locking script to take the transaction fee into consideration. After step 3, the top element on the stack is the value from the previous output. By adding `<TxFee>` `OP_SUB` before the concatenation in step 4, Alice allows Bob to pay the transaction fee from the previous output. This will lead to diminishing value of the output over spends, which can act as a desired feature as it sets the total number of spends Bob are entitled to.

## 1.4 Optimisation

As  $Data_1$  contains  $Data_2$ , we can construct  $Data_2$  from  $Data_1$ . In other words, we assume that the current output is identical to the previous output and use the previous output to construct the message. If it passes `OP_CHECKSIG`, then the two outputs must be identical. The script segment of `[outputsRequest]` can be re-written as:

```
[outputsRequest]:= OP_2DUP OP_CAT OP_TOALTSTACK OP_SWAP OP_CAT OP_HASH256
<item 8> OP_SWAP OP_CAT OP_FROMALTSTACK OP_SWAP OP_CAT OP_CAT <item 10
and 11> OP_CAT
```

Input data: `<item 1 to 4> <item 5 and 6> <item 7>`

With this new `[outputsRequest]`, we can update the locking script in  $TxID_0$  and  $TxID_1$  as:

```
[outputsRequest][sign] OP_CHECKSIGVERIFY OP_DUP OP_HASH160 <H(PK_B)>
OP_EQUALVERIFY OP_CHECKSIG
```

and the unlocking script as:

```
<SigB> <PKB> <Data1> <Data2> <Data3>
```

where  $Data_1$  is item 1 to 4:

```
010000002268f59280bdb73a24aae224a0b30c1f60b8a386813d63214f86b98261a6b8763bb
13029ce7b1f559ef5e747fcac439f1455a2ec7c5f09b72290795e70665044TxID000000000
```

$Data_2$  is item 5 and 6:

```
011b{[outputsRequest] [sign] OP_CHECKSIGVERIFY OP_SWAP <0x68> OP_SPLIT
OP_NIP OP_SWAP OP_8 OP_SPLIT OP_SWAP OP_CAT OP_EQUALVERIFY OP_DUP
OP_HASH160 <H(PK_B)> OP_EQUALVERIFY OP_CHECKSIG}
```

$Data_3$  is item 7: `e803000000000000`.

The size of  $TxID_1$  is 941 bytes. A step-by-step execution is given below, where Step 1 to 5 is from `[outputsRequest]`.

**Table 4** - Stack execution with optimisation

Step	The stacks	To execute
1	<code>&lt;Sig<sub>B</sub>&gt; &lt;PK<sub>B</sub>&gt; &lt;Data<sub>1</sub>&gt; &lt;Data<sub>2</sub>&gt; &lt;Data<sub>3</sub>&gt;</code>	<code>OP_2DUP OP_CAT OP_TOALTSTACK</code>

<b>2</b>	$\langle Sig_B \rangle \langle PK_B \rangle \langle Data_1 \rangle \langle Data_2 \rangle \langle Data_3 \rangle$ ALTSTACK: $\langle \text{item 5 to 7} \rangle$	OP_SWAP OP_CAT OP_HASH256
<b>3</b>	$\langle Sig_B \rangle \langle PK_B \rangle \langle Data_1 \rangle \langle \text{item 9} \rangle$ ALTSTACK: $\langle \text{item 5 to 7} \rangle$	$\langle \text{item 8} \rangle$ OP_SWAP OP_CAT
<b>4</b>	$\langle Sig_B \rangle \langle PK_B \rangle \langle Data_1 \rangle \langle \text{item 8 and 9} \rangle$ ALTSTACK: $\langle \text{item 5 to 7} \rangle$	OP_FROMALTSTACK OP_SWAP OP_CAT
<b>5</b>	$\langle Sig_B \rangle \langle PK_B \rangle \langle Data_1 \rangle \langle \text{item 5 to 9} \rangle$	OP_CAT $\langle \text{item 10 and 11} \rangle$ OP_CAT
<b>6</b>	$\langle Sig_B \rangle \langle PK_B \rangle \langle \text{item 1 to 11} \rangle$	[sign] OP_CHECKSIGVERIFY
<b>7</b>	$\langle Sig_B \rangle \langle PK_B \rangle$	OP_DUP OP_HASH160 $\langle H(PK_B) \rangle$ OP_EQUALVERIFY OP_CHECKSIG
<b>8</b>	True	

Further improvement can be made by using the alt stack for storing  $G_x$  and  $n$ . Each of them is of size 32 bytes. As  $G_{compress}$  is  $G_x$  and  $\frac{n}{2}$  can be derived from  $n$ , we can use several opcodes to reference them from the alt stack. However, it would introduce complications when designing the script. For example, a comparison of two versions of [sign] is shown below.

Before:

```
[sign]:= OP_HASH256  $G_x$  OP_ADD  $n$  OP_MOD [toDER] SIGHASH_FLAG OP_CAT  $G_x$ 

[toDER]:= [toCanonical][concatenations]

[toCanonical]:= OP_DUP  $n/2$  OP_GREATERTHAN OP_IF  $n$  OP_SWAP OP_SUB OP_ENDIF

[concatenations]:= OP_SIZE OP_DUP <0x24> OP_ADD <0x30> OP_SWAP OP_CAT
<0220|| $G_x$ > OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
```

After:

```
[sign]:= OP_HASH256 Gx OP_DUP OP_TOALTSTACK OP_ADD n OP_DUP OP_TOALTSTACK
OP_MOD [toDER] SIGHASH_FLAG OP_CAT OP_FROMALTSTACK

[toDER]:= [toCanonical][concatenations]

[toCanonical]:= OP_DUP OP_FROMALTSTACK OP_DUP OP_TOALTSTACK OP_2 OP_DIV
OP_GREATERTHAN OP_IF OP_FROMALTSTACK OP_SWAP OP_SUB OP_ENDIF

[concatenations]:= OP_SIZE OP_DUP <0x24> OP_ADD <0x30> OP_SWAP OP_CAT
<0220> OP_FROMALTSTACK OP_DUP OP_TOALTSTACK OP_CAT OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT
```

We added 15 opcodes and removed two instances of  $G_x$  and two instances of  $n$ . The total saving is  $(32 \times 2 + 32 \times 2) - 15 = 113$  bytes. Therefore, the size of the spending transaction  $TxD_1$  can be further reduced to 828 bytes.

It is safe to say that with all available optimisations, the size of a spending transaction can be around 1KB.

Note that any example scripts shown above are not meant to be used in mainnet transactions. They are written for illustration purpose. Some simplifications are implicitly assumed. For example, reversing endianness is omitted. For a more practical and comprehensive guide, please refer to Appendix.

## 2 Security Analysis

In this section, we offer some security proofs and analysis around PUSHTX. This is considered a formal approach to PUSHTX and some of the insights we described in the previous section.

First of all, we notice that there are effectively two signatures in the script execution. One is created by PUSHTX technique, the other one is provided in the unlocking script for P2PKH. The first signature provides data integrity and enables enforcement on the spending transaction, while the second provides authenticity and ensure that only the intended recipient can create the spending transaction<sup>2</sup>.

**Claim 1:** If  $(r, s)$  is a valid ECDSA signature with respect to a public key  $P$  on the messages  $m$ , then it is computationally infeasible to construct  $m'$  such that  $m' \neq m$  and  $(r, s)$  is still valid on  $m'$  with respect to  $P$ , assuming that the cryptographic hash function used is pre-image resistant and collision resistant.

**Proof:**

let  $z = \text{hash}(m)$  and  $z' = \text{hash}(m')$  for some  $m'$ .

Let  $u = zs^{-1} \bmod n$ ,  $u' = z's^{-1} \bmod n$ ,  $v = rs^{-1} \bmod n$ , and  $R = kG$  where  $R_x = r \bmod n$ .

Also let  $R'$  denote the points such that  $R' \neq R$  and  $R'_x = r \bmod n$ .

Note that<sup>3</sup>  $R'$  can either be  $-R$  or one of the two points,  $(x, \pm y)$ ,

where  $x = R_x + n$  if  $R_x < n$  and  $x = R_x - n$  if  $R_x > n$ .

So, from the signature verification, we have  $[uG + vP]_x = [u'G + vP]_x = r \bmod n$ .

There are two cases to consider.

Case 1:  $uG + vP = u'G + vP = R$

$$\Rightarrow u = u' \bmod n$$

$$\Rightarrow z = z' \bmod n$$

$$\Rightarrow m = m' \text{ under the assumption that the hash function is collision resistant.}$$

Case 2:  $uG + vP = R$  and  $u'G + vP = R'$

$$\Rightarrow u'G = R' - vP$$

$$\Rightarrow \text{Given that the message } m, \text{ the public } P, \text{ and the value } r \text{ are all fixed, there are maximum three different values that } u' \text{ can take, each of which corresponds to a value of } R'.$$

---

<sup>2</sup> In some use cases where authenticity is not required and it is desired to have anyone being able to create a spending transaction, the script segment P2PKH can be omitted. However, any implications and risks should be analysed and assessed before such approach.

<sup>3</sup> On secp256k1 curve, the group order  $n$  is less than the curve modulo  $p$ , but they are of the same bit length. Therefore, an equation  $x = a \bmod n$  will have maximum two solutions for  $x \in [0, p - 1]$ . The maximum number of solutions can only be achieved when  $a \in [0, p - n)$ .

- ⇒  $z' = u's$  also has maximum three values.
- ⇒ Finding  $m'$  such that  $hash(m') = z'$  where  $z'$  can only be one of the three fixed values is computationally infeasible assuming that the hash function is pre-image resistant.

Therefore, we can conclude that it is computationally infeasible to find  $m'$  such that  $m' \neq m$  and  $(r, s)$  is still valid on  $m'$  with respect to  $P$ .

Note that Claim 1 implies that PUSHTX technic is secure to use if we assume that double SHA256 is preimage resistant and collision resistant. Moreover, it implies that the data integrity is preserved even when we sacrifice authenticity in ECDSA by making both the private key and the ephemeral key available to the public.

**Claim 2:** Public key  $P$  must be fixed in the locking script.

**Reasoning:**

Suppose  $P$  is not fixed and  $(r, s)$  is a valid signature with respect to  $P$  on  $m$ .

Let  $z' = hash(m')$ .  $u' = z's^{-1}$ , and  $v = rs^{-1}$

We want to find  $P'$  such that  $u'G + vP' = R$

$$P' = v^{-1}(R - u'G)$$

Now  $(r, s)$  is valid with respect to  $P'$  on  $m'$ .

Therefore  $P$  must be fixed in the locking script.

**Claim 3:**  $k$  should be fixed in the locking script.

**Reasoning:**

Suppose  $(r, s)$  is a valid signature generated in the locking script with respect to  $P$  on  $m$ .

Suppose  $k$  is not fixed in the locking script and is provided in the unlocking script.

Then an adversary can:

1. intercept the spending transaction, and
2. replace  $k$  with  $k'$  in the unlocking script.

Then  $(r', s')$  generated in the locking script will be a valid signature with respect to  $P$  on  $m$ .

Transaction will still be valid, but the transaction ID is changed.

**Claim 4:** Sighash flag should be fixed in the locking script.

**Reasoning:**

Suppose  $(r, s)$  is a valid signature generated in the locking script with respect to  $P$  on  $m$ .

Suppose sighash flag is not fixed in the locking script and is provided in the unlocking script.

1. Intercept the spending transaction



2. Change the sighash flag
3. Update the message  $m$  accordingly to  $m'$ .

In some use case, this would invalidate the transaction. E.g., the locking script expects multiple inputs and outputs with sighash flag “ALL”; changing the flag to anything else would invalidate the script execution.

In others, this would change the transaction ID without invalidating the transaction. E.g., the locking script only enforces conditions on the outputs of its spending transaction; adding or removing “ANYONECANPAY” would not invalidate the transaction, but will change the transaction ID.





```

"vin": [
  {
    "txid":
"52685bdbaae5c76887c23cee699bc48f293192a313c19b9fad4c77b993655df5",
    "vout": 0,
    "scriptSig": {
      "asm":
"3044022079be667ef9dcbbac55a06295ce870b07029bfcdb2dce28d959f2815b16f81798
02201229c3605c61c4133b282cc30ece9e7d5c3693bf2cd1c03a3caadcd9f25900a5[ALL|
FORKID]
0279be667ef9dcbbac55a06295ce870b07029bfcdb2dce28d959f2815b16f81798",
      "hex":
"473044022079be667ef9dcbbac55a06295ce870b07029bfcdb2dce28d959f2815b16f817
9802201229c3605c61c4133b282cc30ece9e7d5c3693bf2cd1c03a3caadcd9f25900a5412
10279be667ef9dcbbac55a06295ce870b07029bfcdb2dce28d959f2815b16f81798"
    },
    "sequence": 4294967295
  }
],
"vout": [
  {
    "value": 49.99999388,
    "n": 0,
    "scriptPubKey": {
      "asm": "OP_2DUP OP_BIN2NUM 512 OP_SUB 8 OP_NUM2BIN OP_SWAP OP_CAT
OP_HASH256 -2147483647 OP_SWAP OP_CAT OP_CAT OP_CAT OP_CAT
0000000041000000 OP_CAT OP_HASH256 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT 0 OP_CAT OP_BIN2NUM
9817f8165b81f259d928ce2ddbfc9b02070b87ce9562a055acbbdcf97e66be79 OP_ADD
414136d08c5ed2bf3ba048afe6dcaebafefefffffffffffffffffffffffffffffffffff00 OP_MOD
OP_DUP 414136d08c5ed2bf3ba048afe6dcaebafefefffffffffffffffffffffffffffffffffff00
2 OP_DIV OP_GREATERTHAN OP_IF
414136d08c5ed2bf3ba048afe6dcaebafefefffffffffffffffffffffffffffffffffff00
OP_SWAP OP_SUB OP_ENDIF OP_SIZE OP_DUP OP_TOALTSTACK OP_TOALTSTACK 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT OP_FROMALTSTACK 32 OP_GREATERTHAN OP_IF 1 OP_SPLIT OP_ENDIF

```





```
000000fb472de1f838d9560dc7b19b1ab62b0c6ed60580779017d3cd32d22bcc051ce13bb
13029ce7b1f559ef5e747fcac439f1455a2ec7c5f09b72290795e7066504434ca0c37a8bc
b75db331e9fbc34b976fd93738a70cf883b264c0a40111d4b988000000004d3502fd32026
e810200029458807c7eaa04fffffffff7c7e7e7e7e0800000000410000007eaa517f517f51
7f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517
f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f
7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7
c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c
01007e81209817f8165b81f259d928ce2ddbfc9b02070b87ce9562
a055acbbdcf97e66be799321414136d08c5ed2bf3ba048afe6dcaebafefeffffffffffffff
ffffffffffffffff00977621414136d08c5ed2bf3ba048afe6dcaebafefeffffffffffffff
ffffffffffffffff005296a06321414136d08c5ed2bf3ba048afe6dcaebafefeffffffffff
ffffffffffffffff007c946882766b6b517f517f517f517f517f517f517f517f517f517f
517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f5
17f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f517f51
7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7e7c7
e7c7e6c0120a0637c7e68827601249301307c7e23022079be667ef9dcbba55a06295ce87
0b07029bfcdb2dce28d959f2815b16f81798027e7c7e7c7e01417e210279be667ef9dcbba
c55a06295ce870b07029bfcdb2dce28d959f2815b16f81798ad76a914751e76e8199196d4
54941c45d1b3a323f1433bd688ac089cef052a01000000"
```

```
    },
    "sequence": 4294967295
  }
],
"vout": [
  {
    "value": 49.99998876,
    "n": 0,
    "scriptPubKey": {
      "asm": "OP_2DUP OP_BIN2NUM 512 OP_SUB 8 OP_NUM2BIN OP_SWAP OP_CAT
OP_HASH256 -2147483647 OP_SWAP OP_CAT OP_CAT OP_CAT OP_CAT
0000000041000000 OP_CAT OP_HASH256 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP OP_CAT OP_SWAP
OP_SWAP OP_CAT OP_SWAP OP_CAT 0 OP_CAT OP_BIN2NUM
9817f8165b81f259d928ce2ddbfc9b02070b87ce9562a055acbbdcf97e66be79 OP_ADD
414136d08c5ed2bf3ba048afe6dcaebafefeffffffffffffffffffffffffffffff00 OP_MOD
OP_DUP 414136d08c5ed2bf3ba048afe6dcaebafefeffffffffffffffffffffffffffffff00
2 OP_DIV OP_GREATERTHAN OP_IF
414136d08c5ed2bf3ba048afe6dcaebafefeffffffffffffffffffffffffffffff00
OP_SWAP OP_SUB OP_ENDIF OP_SIZE OP_DUP OP_TOALTSTACK OP_TOALTSTACK 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1
OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1 OP_SPLIT 1"
```





